

9.2 Auswahl einfacher Funktionen

Beispiele für einfache Funktionen

Die untere Tabelle zeigt beispielhaft die Anwendung verschiedener, einfacher Funktionen. Die Berechnung der Funktionen in der Spalte *Ergebnis* basiert auf den Werten des Bereichs B2:F2 bzw. B3:F3 der rechten Tabelle.

	A	B	C	D	E	F
1						
2	Zahlen	45	32	13	54	4
3	Werte	5 Bananen			22 Äpfel	
4						

Beispielwerte für die unteren Funktionsberechnungen

Syntax der Funktion	Bedeutung	Beispiel	Ergebnis
SUMME(Zahl1;Zahl2;...)	Summe	=SUMME(B2:F2)	148
MITTELWERT(Zahl1;Zahl2;...)	Mittelwert/Durchschnitt	=MITTELWERT(B2:F2)	29,6
MIN(Zahl1;Zahl2;...)	Kleinster Wert	=MIN(B2:F2)	4
MAX(Zahl1;Zahl2;...)	Größter Wert	=MAX(B2:F2)	54
ANZAHL(Wert1;Wert2;...)	Anzahl der Zahlen in der Argumentenliste	=ANZAHL(B3:F3)	2
ANZAHL2(Wert1;Wert2;...)	Anzahl der Werte (= beliebige Daten) in der Argumentenliste	=ANZAHL2(B3:F3)	4

Die Funktion ANZAHL2 liefert die Anzahl aller Einträge (Zahlen, Texte etc.) im markierten Zellbereich, während die Funktion ANZAHL lediglich die Anzahl der numerischen Einträge (Zahlen) berechnet.

Einfache Funktionen mit dem Symbol AUTOSUMME einfügen

Die Funktion SUMME können Sie direkt durch Anklicken des Symbols Σ (Register START, Gruppe BEARBEITEN) oder über die Schaltfläche AUTOSUMME (Register FORMELN, Gruppe FUNKTIONSBIBLIOTHEK) einfügen. Weitere Funktionen stehen Ihnen über das geöffnete Listenfeld des Symbols bzw. der Schaltfläche zur Verfügung.

- ▶ Markieren Sie die Zelle, in die das Ergebnis der gewünschten Funktion eingetragen werden soll.
- ▶ Klicken Sie im Register FORMELN in der Gruppe FUNKTIONSBIBLIOTHEK auf den Pfeil der Schaltfläche AUTOSUMME, um weitere Funktionen anzeigen zu lassen.
- ▶ Wählen Sie im Listenfeld eine Funktion aus, z. B. MITTELWERT. Die Formel der Funktion wird in der Ergebniszelle eingeblendet. Unterhalb der Ergebniszelle erscheint automatisch die Beschreibung der Funktion in einer Infobox, in der das aktuelle Argument fett formatiert ist.
- ▶ Bearbeiten Sie die Argumentenliste bei Bedarf.
- ▶ Bestätigen Sie die Eingaben mit der **RETURN** \leftarrow -Taste.

Eingabe der Funktion MITTELWERT über das Listenfeld der Schaltfläche AUTOSUMME

- ✓ Excel schlägt standardmäßig in der eingefügten Funktion einen Zellbereich vor. Diesen können Sie bei Bedarf korrigieren, indem Sie einen anderen Zellbereich markieren.
- ✓ Wollen Sie zusätzlich weitere Zellbereiche eingeben, können Sie die Argumente wie gewohnt durch die Eingabe des Trennzeichens Semikolon an der aktuellen Cursorposition ① trennen. Nach der Eingabe des Semikolons wird in der eingeblendeten Infobox automatisch das zweite Argument fett formatiert und die Beschreibung um ein drittes mögliches Argument erweitert.
- ✓ Hilfe zur Funktion erhalten Sie, indem Sie den Link *Funktionsname* ② in der Infobox anklicken.

Die Funktion RUNDEN

Syntax der Funktion:

RUNDEN(Zahl;Anzahl_Stellen)

Die Funktion RUNDEN rundet den Wert *Zahl* auf die im Argument *Anzahl_Stellen* angegebene Dezimalstelle. Excel wendet dabei die Regeln des kaufmännischen Rundens an:

- ✓ Bei einer Ziffer, die kleiner als **fünf** ist, wird abgerundet.
- ✓ Ziffern, die größer oder gleich fünf sind, werden aufgerundet.
- ✓ Ein **negativer** Wert im Argument *Anzahl_Stellen* bewirkt ein Runden an der entsprechenden Stelle **vor** dem Komma.

Beispiel	Ergebnis
=RUNDEN(7,1213;1)	7,1
=RUNDEN(-31,446;2)	-31,45
=RUNDEN(31,446;2)	31,45
=RUNDEN(31,446;0)	31
=RUNDEN(1273;-2)	1300

Hinweise zum Runden von Zahlen

- ✓ Über die Zellformatierung können Sie Werte mit Dezimalstellen als gerundete Werte anzeigen lassen. Beachten Sie, dass die Zellformatierung lediglich die Anzeige der Werte bestimmt. Die Berechnung erfolgt dagegen mit dem genauen Inhalt der Zelle.
- ✓ Die Funktionen ABRUNDEN und AUFRUNDEN benötigen die gleichen Argumente wie die Funktion RUNDEN. Mit diesen Funktionen können Sie festlegen, ob eine Zahl an der angegebenen Stelle generell abgerundet bzw. aufgerundet werden soll.
- ✓ Wollen Sie eine Dezimalzahl auf die nächste ganze Zahl (Integer) abrunden, können Sie die Funktion GANZZAHL(*Zahl*) verwenden. Die Funktion GANZZAHL(31,446) liefert beispielsweise als Ergebnis 31.

Anzeige möglicher Fehlerwerte innerhalb von Zellen

Fehlerwerte Mögliche Ursachen und Methoden zur Fehlerkorrektur

- #####** Dieser Fehlerwert wird beispielsweise angezeigt, wenn der Zellinhalt durch eine zu geringe Spaltenbreite nicht korrekt angezeigt werden kann. Stellen Sie die optimale Spaltenbreite ein.
- #BEZUG!** Überprüfen Sie Zellbezüge, die sich eventuell durch Löschen von Zellen geändert haben.
- #DIV/0!** Excel hat versucht, durch 0 zu dividieren. Überprüfen Sie (beispielsweise mit der Formelüberwachung), ob die Formel mithilfe der Zellbezüge auf die richtigen Zellen zugreift.
- #NAME?** Namen von Zellen oder Zellbereichen wurden falsch geschrieben oder existieren nicht (vgl. Kapitel 10). Überprüfen Sie die Schreibweise der Namen, die Sie vergeben haben.
- #NULL!** Die Zellbezüge können nicht gefunden werden. Überprüfen Sie die Schreibweise der Argumente auf Korrektheit.
- #NV** Die Formeln enthalten möglicherweise Bezüge auf leere Zellen. Überprüfen Sie Vorgänger- und Nachfolgerzellen mithilfe der Formelüberwachung.
- #WERT!** Der Datentyp des Arguments stimmt nicht mit der erforderlichen Syntax überein. Überprüfen Sie die Argumente der Formel.
- #ZAHL!** Überprüfen Sie, ob die Argumente in den korrekten Zahlenbereichen liegen.

- ① Öffnen Sie die Datei *Immobilien*.
- ② Erstellen Sie in Zelle C5 eine Formel, die den Provisionssatz aus Zelle E10 zum Angebotsbetrag hinzuaddiert.
- ③ Erzeugen Sie mithilfe der Ausfüllfunktion die restlichen Ergebnisse der Spalte C.
- ④ Berechnen Sie in Spalte E das höchste, niedrigste und durchschnittliche Angebot, und führen Sie in Spalte F die gleichen Berechnungen für die Angebote inklusive Provision durch. Verwenden Sie hierfür das Symbol Σ .
- ⑤ Ermitteln Sie die Anzahl der Angebote in Zelle E8 mithilfe des Funktions-Assistenten.

Zellen und Bereiche benennen

- ▶ Markieren Sie die Zelle oder den Bereich ①, die bzw. der benannt werden soll.
- ▶ Klicken Sie in das Namenfeld ② an der linken Seite der Bearbeitungsleiste, und geben Sie den gewünschten Namen ein.
- ▶ Bestätigen Sie mit **RETURN**.

Kosten		5000
	A	B
1		Einzelkosten
2	Entwicklung	5.000,00 €
3	Produktion	15.000,00 €
4	Werbung	① 2.500,00 €
5	Sonstiges	1.500,00 €

Zellbereich benennen

Einen Namen können Sie einer einzelnen Zelle, mehreren unabhängigen Zellen oder einem zusammenhängenden Zellbereich zuweisen. Ebenso können Sie für Spalten und Zeilen Spalten- bzw. Zeilenamen vergeben.

Richtlinien für Namen

- ✓ Der Name muss mit einem Buchstaben, dem Unterstrich **_** oder dem Zeichen **** beginnen.
- ✓ Ein Name darf aus bis zu 255 Zeichen bestehen und kann Buchstaben, Zahlen, die Zeichen **_**, ****, **?** und den Punkt **.** enthalten.
- ✓ Leerzeichen, Semikola, Bindestriche oder Doppelpunkte in Namen sind nicht erlaubt. Verwenden Sie stattdessen einen Unterstrich (z. B. Umsatz_2006) oder einen Punkt.
- ✓ Namen, die Zellbezügen ähnlich sind (beispielsweise B23), sind nicht erlaubt.
- ✓ Bei Namen wird nicht zwischen Groß- und Kleinschreibung unterschieden.

Namen automatisch vergeben

Sie können Spalten- bzw. Zeilenüberschriften im Tabellenblatt auch automatisch als Namen für die entsprechenden Zellen übernehmen lassen. Die Verwendung von Überschriften als Namen erleichtert die Lesbarkeit von Formeln.

- ▶ Markieren Sie die Zellen, die Sie benennen möchten, inklusive der Zeilen- bzw. Spaltenbeschriftungen ①.
- ▶ Klicken Sie im Register FORMELN in der Gruppe DEFINIERTE NAMEN auf die Schaltfläche AUSWAHL ERSTELLEN.
Alternative: **STRG** **⇧** **UMSCHALTEN** **F3**
- ▶ Bestimmen Sie anhand der Kontrollfelder im Bereich ②, aus welchen Zellen innerhalb der Markierung die Namen übernommen werden sollen, falls das entsprechende Kontrollfeld nicht bereits automatisch aktiviert ist.

Spaltenüberschriften als Namen festlegen

Wollen Sie sowohl Zeilen- als auch Spaltenüberschriften übernehmen, markieren Sie den kompletten Bereich ③ und aktivieren Sie die entsprechenden Kontrollfelder im Dialogfenster NAMEN ERSTELLEN. Dem Bereich B4:B6 wird beispielsweise der Name *Einzelpreis*, dem Bereich B4:C4 der Name *Bleistift* zugewiesen.

	A	B	C	D	E
1	Preisberechnung				
2					
3		Einzelpreis	Menge	Preis	
4	Bleistift	0,45 €	8		
5	Fuller	③ 7,99 €	2		
6	Tintenkiller	0,89 €	5		

Zeilen- und Spaltenüberschriften als Namen festlegen

11.1 Aufbau und Einsatzgebiete der Funktion WENN

Die Funktion WENN einsetzen

Mit der Funktion WENN können Sie die Ausführung einer Formel von dem Eintreten einer Bedingung abhängig machen. Je nachdem, ob diese wahr oder falsch ist, liefert die Funktion WENN unterschiedliche Ergebnisse.



Die WENN-Funktion finden Sie im Register FORMELN in der Gruppe FUNKTIONSBIBLIOTHEK in der Liste der Schaltfläche LOGISCH oder im Funktions-Assistenten unter der Kategorie LOGIK.

In der nebenstehenden Tabelle werden die Provisionen für die Mitarbeiter in Abhängigkeit vom erzielten Umsatz berechnet. Es werden zwei unterschiedliche Provisionssätze gezahlt: Beträgt der Umsatz weniger als 300.000 €, werden 4 % Provision gezahlt, anderenfalls 8 %.


Für die Formel gilt allgemein:

Syntax der Funktion: `WENN(Prüfung;Dann_Wert;Sonst_Wert)`

Beispiellösung (in Zelle C4): `=WENN(B4<300000;B4*4%;B4*8%)`

	A	B	C
1	Provisionsberechnung		
2			
3	Name	Umsatz	Provision
4	Blohme	250.000 €	10.000 €
5	Schulz	350.000 €	28.000 €
6	Müller	300.000 €	24.000 €

Provisionen berechnen



Die Argumente der Funktion WENN

Ist die Bedingung erfüllt bzw. wahr, wird der **Dann_Wert** ausgewertet. Ist die Bedingung nicht erfüllt bzw. falsch, wird der **Sonst_Wert** ausgewertet.

Die Argumente der Funktion WENN werden in der folgenden Tabelle beschrieben:

Prüfung (Bedingung) Zur Formulierung einer Bedingung werden zwei Werte miteinander verglichen. Dafür sind sogenannte Vergleichsoperatoren erforderlich: = (gleich), <> (ungleich), < (kleiner), > (größer), <= (kleiner oder gleich), >= (größer oder gleich).
Das Ergebnis einer Prüfung wird mit den Wahrheitswerten *Wahr* und *Falsch* ausgedrückt. Wird zur Prüfung nur ein Zellbezug ohne Operator angegeben, ergibt die Prüfung den Wahrheitswert *Wahr*, wenn sich in der Zelle ein Wert ungleich null befindet. Der Wahrheitswert *Falsch* wird angenommen, wenn die Zelle leer ist oder den Wert null enthält.

Dann_Wert Wenn die Bedingung wahr ist, wird das Argument *Dann_Wert* ausgeführt. Dabei gilt:
 ✓ Das Argument kann ein Wert oder eine Formel (Funktion) sein.
 ✓ Wird als Wert ein Text in die Zelle geschrieben, muss dieser in Anführungszeichen gesetzt werden (z. B. "keine Provision").
 ✓ Soll das Ergebnis eine leere Zelle sein, sind zwei Anführungszeichen direkt hintereinander zu setzen ("").

Sonst_Wert Wenn die Bedingung falsch ist, wird das Argument *Sonst_Wert* ausgeführt. Es gelten die gleichen Regeln wie für das Argument *Dann_Wert*.

Weitere Beispiele für die Verwendung der Funktion WENN

Beispielformeln

`=WENN(A1<>0;B1/A1;0)`

`=WENN(Gewinn>0;"Gewinn";"Verlust")`

Bedeutung

Wenn der Zellinhalt A1 ungleich 0 ist, dann wird B1 durch A1 dividiert, ansonsten wird 0 ausgegeben.

Wenn der Inhalt der Zelle *Gewinn* größer 0 ist, dann wird der Text *Gewinn* ausgegeben, anderenfalls *Verlust*.

11.2 Verschachtelte WENN-Funktionen einsetzen

WENN-Funktionen ineinander verschachteln

Manche Problemstellungen erfordern mehrere Bedingungen. In solchen Fällen müssen WENN-Funktionen ineinander verschachtelt werden.

Zur Berechnung der nebenstehend abgebildeten Provisionen werden beispielsweise folgende Bedingungen aufgestellt:


	A	B	C
1	Provisionsberechnung		
2			
3	Name	Umsatz	Provision
4	Blohme	250.000 €	25.000 €
5	Schulz	150.000 €	9.000 €
6	Müller	30.000 €	0 €

- WENN** Umsatz kleiner 50.000 € → **DANN** keine Provision
- **SONST WENN** Umsatz kleiner 200.000 € → **DANN** 6 % des Umsatzes
- **SONST** 10 % des Umsatzes

Beispiellösung (in Zelle C4): =WENN(B4<50000;0;WENN(B4<200000;B4*6%;B4*10%))

- ✓ Die äußere Funktion WENN stellt die Bedingung $B4 < 50000$. Ist die Bedingung erfüllt, wird eine 0 ausgegeben.
- ✓ Wenn die Bedingung $B4 < 50000$ nicht erfüllt ist, wird das Argument *Sonst_Wert* durchgeführt, das aus einer weiteren Funktion WENN besteht.
- ✓ Die zweite Funktion WENN stellt die Bedingung $B4 < 200000$.

Zwei verschiedene Ergebnisse erfordern eine Bedingung (eine Funktion WENN); drei mögliche Ergebnisse erfordern zwei Bedingungen (zwei verschachtelte WENN-Funktionen). Sie können bis zu 64 WENN-Funktionen ineinander verschachteln.



Insbesondere bei verschachtelten Funktionen ist die Übersichtlichkeit durch die Verwendung von Zellbezügen stark eingeschränkt. Versehen Sie daher zur Vereinfachung Zellen und Zellbereiche mit Namen.

Bedingungen verknüpfen

Mithilfe der logischen Funktionen UND, ODER und NICHT lassen sich mehrere Bedingungen miteinander verknüpfen. Die Funktionen werden als Argument *Prüfung* in die Funktion WENN eingefügt.

UND(Wahrheitswert1; Wahrheitswert2;...) Diese Funktion liefert das Ergebnis *Wahr*, wenn alle Bedingungen (maximal 255) wahr sind.

Beispiel: Wenn der Umsatz zwischen 20.000 und 30.000 liegt, soll in die Zelle das Wort *Ok*, ansonsten *Nachfragen* eingetragen werden.

=WENN(UND(Umsatz>=20000;Umsatz<=30000);"Ok";"Nachfragen")

ODER(Wahrheitswert1; Wahrheitswert2;...) Diese Funktion liefert das Ergebnis *Wahr*, wenn mindestens eine der Bedingungen (maximal 255) *Wahr* ist.

Beispiel: Wenn in Zelle A8 nichts oder in Zelle A10 eine 0 eingetragen ist, dann soll die Meldung *Keine Werte eingetragen* ausgegeben werden.

=WENN(ODER(A8="";A10=0);"Keine Werte eingetragen";"")

NICHT(Wahrheitswert) Die Funktion NICHT kehrt den Wahrheitswert der Bedingung um. Ergibt die Bedingung den Wahrheitswert *Wahr*, ist das Ergebnis der Funktion *Falsch* und umgekehrt.

Beispiel: Wenn der Umsatz nicht größer als 20.000 ist und nicht innerhalb von 8 Wochen erzielt wurde, soll eine Provision von 5 %, anderenfalls von 10 % berechnet werden.

=WENN(UND(NICHT(Umsatz>20000);NICHT(Zeit<=8));Umsatz*5%;Umsatz*10%)

9.4 Verschachtelte Funktionen erstellen

Verschachtelte Funktionen mit dem Assistenten erstellen

Als Argumente einer Funktion können Sie auch das Ergebnis einer anderen Funktion, einer sogenannten verschachtelten Funktion, verwenden.


Sie können z. B. den Mittelwert des Bereichs B2:B5 berechnen und als Argument für die Funktion RUNDEN verwenden, wenn Sie mit dem gerundeten Wert weiterarbeiten möchten. Die Funktion MITTELWERT fungiert als verschachtelte Funktion innerhalb der Funktion RUNDEN.

	A	B	C	D
1	Klassenstufe	Anzahl Kinder		
2	1. Klasse	75		
3	2. Klasse	89		
4	3. Klasse	62		
5	4. Klasse	71		
6	Mittelwert (gerundet)	74		
7				


Verschachtelte Funktion MITTELWERT

Diese Berechnungen lassen sich in einer einzigen Formel durchführen. Wichtig für die korrekte Berechnung ist die richtige Reihenfolge der Funktionen: Beschreiben Sie zuerst die Funktion, die zuletzt ausgeführt werden soll:


1. Der Wert soll gerundet werden, z. B. auf eine ganze Zahl.
2. Der Mittelwert aus dem Bereich B2:B5 soll berechnet werden und anschließend auf eine ganze Zahl gerundet werden.




Verschachtelte Funktion eingeben

- ▶ Markieren Sie die Zelle, in die das Ergebnis eingetragen werden soll.
- ▶ Klicken Sie auf das Symbol  in der Bearbeitungsleiste, um den Funktions-Assistenten zu starten.
- ▶ Wählen Sie die gewünschte Funktion (hier: RUNDEN).
- ▶ Setzen Sie im Funktions-Assistenten den Cursor in das Eingabefeld des Argumentes, das mit einer weiteren Funktion berechnet werden soll (hier: Argument ZAHL ①).
- ▶ Klicken Sie auf das Pfeilsymbol ② in der Bearbeitungsleiste, und wählen Sie im geöffneten Listenfeld FUNKTIONEN eine weitere Funktion (hier: MITTELWERT) aus.

oder Sollte die gewünschte Funktion im Listenfeld nicht angezeigt werden, klicken Sie auf den Eintrag WEITERE FUNKTIONEN und wählen Sie im anschließend erneut geöffneten Funktions-Assistenten die entsprechende Funktion.



Funktionsargumente eingeben



Berechnung durch eine verschachtelte Funktion

Das Dialogfenster FUNKTIONSARGUMENTE ③ wird für die verschachtelte Funktion geöffnet. Die Funktion wird in der Bearbeitungsleiste in die Argumentenliste ④ der ersten Funktion eingefügt.

- ▶ Geben Sie die Argumente für die verschachtelte Funktion ein. Im Beispiel wird automatisch ein Zellbereich ⑤ als Argument für die Funktion MITTELWERT vorgeschlagen.
- ▶ Um wieder zum Dialogfenster FUNKTIONSARGUMENTE der übergeordneten Funktion (hier: RUNDEN) zurückzukehren, klicken Sie innerhalb der Formel in der Bearbeitungsleiste auf den Namen dieser Funktion.
- ▶ Geben Sie hier, falls nötig, weitere Argumente ein (in diesem Beispiel: ANZAHL_STELLEN: 0).
- ▶ Beenden Sie die Formeleingabe mit der Schaltfläche OK.

11.3 Die Funktionen SVERWEIS und WVERWEIS

Einsatzbereich der Funktionen

Das Ergebnis einer Berechnung kann von einer Vielzahl Kriterien abhängig sein, beispielsweise bei einer Provisionsberechnung, die viele Provisionsätze besitzt.

Hier sind die Funktionen SVERWEIS und WVERWEIS hilfreich.

Diese Funktionen suchen anhand bestimmter Kriterien (beispielsweise Umsatz = 290.000 € ①) den zugehörigen Wert (z. B. Provisionsatz = 11 % ②).

	A	B	C	D	E	F	G	H
1	Provisionsberechnung							
2								
3	Name	Umsatz	Provisionsatz	Provision		Provisionstabelle		
4	Blohme	250.000 €	11%	27.500 €		Umsatz	Provisionsatz	
5	Schulz	350.000 €	15%	52.500 €		50.000 €	3%	
6	Müller	300.000 €	13%	39.000 €		100.000 €	5%	
7	Graf	150.000 €	7%	10.500 €		150.000 €	7%	
8	Hohmann	430.000 €	17%	73.100 €		200.000 €	9%	
9	Reich	560.000 €	23%	128.800 €		250.000 €	11%	
10	Klos	① 290.000 €	② 11%	31.900 €		300.000 €	13%	
11	Bieber	470.000 €	19%	89.300 €		350.000 €	15%	
12	Singer	140.000 €	5%	7.000 €		400.000 €	17%	
13	Wagner	48.000 €	#NV	#NV		450.000 €	19%	
14	Erikson	310.000 €	13%	40.300 €		500.000 €	21%	
15	Vogel	29.000 €	#NV	#NV		550.000 €	23%	

Spalte mit Suchkriterien Spalte mit der Funktion SVERWEIS Spaltenindex
Matrix \$F\$5:\$G\$15 (Suchbereich)


Provisionsberechnung mit SVERWEIS

Die Funktionen SVERWEIS und WVERWEIS finden Sie

- ✓ im Register FORMELN in der Gruppe FUNKTIONSBIBLIOTHEK in der Liste des Symbols (NACHSCHLAGEN UND VERWEISEN)
- ✓ im Funktions-Assistenten unter der Kategorie MATRIX

Wie arbeiten die Funktionen?

- ✓ Die Funktionen SVERWEIS und WVERWEIS durchsuchen jeweils die Werte eines bestimmten Tabellenbereichs (Matrix) nach dem gewünschten Suchkriterium (z. B. 290.000 €).
- ✓ Das Suchkriterium wird bei der Funktion SVERWEIS in der ersten Spalte (senkrecht) von oben nach unten gesucht. Bei der Funktion WVERWEIS wird das Suchkriterium in einer zeilenweise angeordneten Matrix in der ersten Zeile (waagrecht) von links nach rechts gesucht.
- ✓ Beim ersten Eintrag, auf den das Suchkriterium zutrifft, wird der zugehörige Wert in der im Argument Spaltenindex/Zeilenindex festgelegten Spalte/Zeile ermittelt und als Ergebnis zurückgegeben. Daher muss die Matrix aufsteigend sortiert sein.
- ✓ Findet die Funktion keinen Wert, der exakt mit dem Suchkriterium übereinstimmt, wird automatisch der nächstkleinere Wert als Suchkriterium genutzt.



Ist das Suchkriterium kleiner als der erste Eintrag der Matrix, erscheint der Fehlerwert #NV. Diesen Fehler können Sie jedoch mit einem sehr kleinen ersten Wert (beispielsweise 0) vermeiden.

Beispiel für die Verwendung der Funktion SVERWEIS

In der zuvor abgebildeten Tabelle Provisionsberechnung wird für jeden Mitarbeiter die Provision gemäß den Werten in der Provisionstabelle errechnet. Hierzu werden die Provisionsätze in Spalte C für die entsprechenden Suchkriterien in Spalte B ermittelt.

Die Provisionstabelle (\$F\$5:\$G\$15) legt die konstante Matrix (den Suchbereich) fest. Die Funktion soll als Ergebnis die Werte der zweiten Spalte (Spaltenindex) des Suchbereichs liefern.

Syntax der Funktion: **SVERWEIS(Suchkriterium;Matrix;Spaltenindex;Bereich_Verweis)**

Beispiellösung (in Zelle C4): =SVERWEIS(B4;\$F\$5:\$G\$15;2;Wahr)

Beispiel für die Verwendung der Funktion WVERWEIS

Die Provisionsberechnung erfolgt aufgrund einer zeilenweise angeordneten Provisionstabelle.

	A	B	C	D	E	F	G	H
1	Provisionstabelle							
2	Umsatz	50.000 €	100.000 €	150.000 €	200.000 €	250.000 €	300.000 €	
3	Provisionsatz	3%	5%	7%	9%	11%	13%	
4								
5								
6	Erzielter Umsatz	180.000 €						
7	Provisionsatz	7%						
8	Provision	12.600 €						

Die Provisionstabelle in der Zeilendarstellung

Syntax der Funktion:

WVERWEIS(Suchkriterium;Matrix;Zeilenindex;Bereich_Verweis)

Beispiellösung (in Zelle B7): =WVERWEIS(B6;\$B\$2:\$G\$3;2;Wahr)

Den Bonus für erfolgreiche Mitarbeiter ermitteln

Übungsdatei: **Bonus**

Ergebnisdatei: **Bonus-E, Bonus-E1, Bonus-E2**

- Erstellen Sie die unten abgebildete Tabelle, oder öffnen Sie die Datei *Bonus*.
- Bei einem Jahresumsatz von mindestens 300.000 € soll der Bonus 2 % betragen. Berechnen Sie in der Spalte *Bonus* den Prozentsatz, den die Filialleiter zusätzlich erhalten. Verwenden Sie für die Berechnungen in Spalte D die Funktion WENN. Benutzen Sie für das Erstellen der Funktion den Funktions-Assistenten.
- Speichern Sie die Datei unter dem Namen *Bonus-E*.

Sie können die Prozentzahl auch direkt in die Formel schreiben. Warum ist es von Vorteil, dass die Prozentzahl in einer gesonderten Zelle steht?

	A	B	C	D	E	F	G	H
1	Bonus für erzielte Umsätze							
2								
3	Filiale	Filialleiter/in	Jahresumsatz	Bonus				
4	Berlin	Fr. Heine	280.000 €	0%				
5	Frankfurt	Hr. Schrenk	549.000 €	2%				
6	Innsbruck	Hr. Wulf	310.000 €	2%				
7	Ludwigshafen	Hr. Böning	250.000 €	0%				
8	München	Fr. Seifert	405.000 €	2%				
9	Zürich	Fr. Warnecke	647.000 €	2%				

Bonusstaffel		
	Betrag	Bonus
unter	300.000 €	0%
ab	300.000 €	2%

Ergebnisdatei "Bonus-E"

- Erweitern Sie die Berechnung: Bei einem Jahresumsatz von mindestens 500.000 € soll der Bonus 3 % betragen. Verwenden Sie für die Verschachtelung der Funktionen ebenfalls den Funktions-Assistenten.
- Speichern Sie die Datei unter dem Namen *Bonus-E1*.

	A	B	C	D	E	F	G	H
1	Bonus für erzielte Umsätze							
2								
3	Filiale	Filialleiter/in	Jahresumsatz	Bonus				
4	Berlin	Fr. Heine	280.000 €	0%				
5	Frankfurt	Hr. Schrenk	549.000 €	3%				
6	Innsbruck	Hr. Wulf	310.000 €	2%				
7	Ludwigshafen	Hr. Böning	250.000 €	0%				
8	München	Fr. Seifert	405.000 €	2%				
9	Zürich	Fr. Warnecke	647.000 €	3%				

Bonusstaffel		
	Betrag	Bonus
unter	300.000 €	0%
ab	300.000 €	2%
ab	500.000 €	3%

Ergebnisdatei "Bonus-E1"


- Verändern Sie die Tabelle gemäß der unteren Abbildung.
- Neuerdings erhalten die Filialleiter nur noch dann einen Bonus, wenn sie neben den bisherigen Kriterien ein bestimmtes Alter erreicht haben. Um beide Bedingungen abzufragen, können Sie die Funktion UND einsetzen. Fügen Sie die Bedingung diesmal manuell in die Funktion WENN ein.
- Speichern Sie die Datei unter dem Namen *Bonus-E2*.


Möglichkeiten der Datums- bzw. Uhrzeiteingaben

Eingabeart	Erläuterung	Beispiele
Datum	Alle Datumseingaben ab dem 1.1.1900 werden als Datumsangaben interpretiert. Bei Datumseingaben kann statt des Punktes zur Trennung auch ein Schräg- (/) bzw. Bindestrich (-) verwendet werden. Excel erkennt automatisch die zweistelligen Jahresangaben 00 - 29 für die Jahre 2000 bis 2029 und 30 - 99 für die Jahre 1930 bis 1999. Alle Jahresangaben vor 1930 bzw. nach 2029 müssen vierstellig angegeben werden.	12.1.07
		12-1-07
		12/1/07
		12.1.2007
		12. Januar 07
Uhrzeit	Bei Uhrzeiteingaben sind Stunden, Minuten und Sekunden jeweils mit einem Doppelpunkt zu trennen. Normalerweise zeigt Excel dann die Zeit im 24-Stunden-Modus an. Wird die Zeitangabe "AM", "A", "PM" oder "P" (in Groß- oder Kleinbuchstaben durch eine Leertaste von der Zeitangabe getrennt) hinzugefügt, stellt Excel die Zeit im amerikanischen 12-Stunden-Modus dar.	12:00
		12:30
		12:30:20
		9 PM
		10 p
Kombinierte Datums- und Zeiteingaben	Kombinierte Datums- und Uhrzeiteingaben werden - durch ein Leerzeichen getrennt - wie die einzelnen Teile eingegeben.	8 am
		12.1.07 12:30
		12 Jan 07 0:30 p

Datums- und Uhrzeitformate zuweisen

Sie können in Excel zahlreiche Formate zur individuellen Gestaltung von Zeitangaben zuweisen.




- ▶ Markieren Sie die Zellen, die Zeitangaben enthalten.
- ▶ Betätigen Sie im Register **START** in der Gruppe **ZAHL** das Symbol . Das Dialogfenster **ZELLEN FORMATIEREN** wird geöffnet und das Register **ZAHLEN** des Dialogfensters ist aktiviert.
- ▶ Wählen Sie die gewünschte Kategorie aus dem Listenfeld ① (**DATUM** bzw. **UHRZEIT**), und aktivieren Sie im Listenfeld ② das entsprechende Format.
- ▶ Im Bereich ③ sehen Sie, wie sich das gewählte Zeitformat auf die erste Zelle des markierten Zellbereichs auswirkt, sofern der markierte Zellbereich Zahlen enthält.
- ▶ Bestätigen Sie mit **Ok**.



Datum und Uhrzeit formatieren

Bei Zeitformaten, die im Listenfeld ② mit einem Sternchen gekennzeichnet sind, richtet sich die Darstellungsart nach den unter Windows in den Regions- und Sprachoptionen festgelegten Formaten.

Mit Sternchen gekennzeichnete Formate können Sie schnell einer Zelle zuweisen, indem Sie im Register **START** in der Gruppe **ZAHL** das Listenfeld öffnen und in der Liste das gewünschte Datumsformat ④ oder ⑤ bzw. das Uhrzeitformat ⑥ wählen. Sofern die aktive Zelle einen entsprechenden Zahlenwert enthält, wird jeweils unterhalb der Formatbezeichnung eine Vorschau angezeigt.

	Datum, kurz ④	02.11.2006
	Datum, lang ⑤	Donnerstag, 2. November 2006
	Zeit ⑥	00:00:00




12.3 Mit Datums- und Uhrzeitfunktionen arbeiten

Argumente von Datums- und Uhrzeitfunktionen

Für Berechnungen mit Datums- und Uhrzeitangaben stellt Excel vielfältige Funktionen zur Verfügung. Die Funktionen finden Sie im Register FORMELN in der Gruppe FUNKTIONSBIBLIOTHEK in der Liste ① der Schaltfläche DATUM UND UHRZEIT bzw. unter der Kategorie DATUM & ZEIT im Funktions-Assistenten.

Erlaubte Argumente für die Datums- und Uhrzeitfunktionen sind

- ✓ die serielle Zahl eines Datums bzw. die Dezimalzahl einer Uhrzeit
- ✓ eine Zeitangabe in Anführungszeichen (Zeittext) bzw. mit vorangestelltem Hochkomma
- ✓ Bezüge auf Zellen, die Zeitangaben enthalten



Datumsfunktionen

HEUTE()	Das heutige Datum
JAHR(Zahl)	Die Jahreszahl bestimmen
MONAT(Zahl)	Die Monatszahl bestimmen
TAG(Zahl)	Den Tag im Monat bestimmen
WOCHENTAG (Zahl;Typ)	Den Wochentag bestimmen, wobei Typ den Wochenanfang festlegt: 1 für Sonntag (1) bis Samstag (7) 2 für Montag (1) bis Sonntag (7) 3 für Montag (0) bis Sonntag (6)

	A	B	C	D
1	Heute	03.11.2006	=HEUTE()	
2				
3	Jahr	2006	=JAHR(B1)	
4	Monat	11	=MONAT(B1)	
5	Tag	3	=TAG(B1)	
6	Wochentag	5	=WOCHENTAG(B1;2)	

Datumsfunktionen verwenden (aktuelles Datum: Freitag, 3. November 2006)

Uhrzeitfunktionen

JETZT()	Das heutige Datum und die aktuelle Zeit
STUNDE(Zahl)	Die Stundenzahl bestimmen
MINUTE(Zahl)	Die Minutenzahl bestimmen
SEKUNDE(Zahl)	Die Sekundenzahl bestimmen

	A	B	C	D
1	Jetzt	03.11.2006 11:58	=JETZT()	
2				
3	Stunde	11	=STUNDE(B1)	
4	Minute	58	=MINUTE(B1)	
5	Sekunde	15	=SEKUNDE(B1)	

Uhrzeitfunktionen verwenden

Weitere Funktionen

TAGE360(Ausgangsdatum; Enddatum;Methode)	Die Anzahl der Tage berechnen (1 kaufmännisches Jahr = 360 Tage)
DATUM(Jahr;Monat;Tag)	Ein vollständiges Datum erzeugen
ZEIT(Stunde;Minute;Sekunde)	Eine vollständige Uhrzeit erzeugen
DATWERT(Datumstext)	Die interne serielle Zahl aus einem Datum bestimmen
ZEITWERT(Zeit)	Die interne Dezimalzahl aus einer Uhrzeit bestimmen

Beispiel: Berechnung über mehr als 24 Stunden durchführen

Sie wollen erfassen, wie lange ein Vorgang bearbeitet wurde. Sie möchten dabei die tägliche Arbeitszeit am Vorgang und die Gesamtarbeitszeit berechnen.

	A	B	C	D	E	F	G
1	Bearbeitung des Vorgangs						
2							
3	Arbeitsbeginn	Pause	Arbeitsende	Zeit pro Tag	Gesamtzeit		
4	08:00	12:30	13:00	17:45	09:15		
5	07:30	13:00	13:15	16:00	08:15		
6	08:00			17:10	09:10		
7						26:40:00	
8							

Zeitberechnungen über einen längeren Zeitraum als 24 Stunden

- ✓ Arbeitsbeginn, Arbeitsende und Pausenzeiten werden in der Tabelle in den Spalten A bis D manuell eingegeben.
- ✓ Zur Ermittlung der täglichen Arbeitszeit wird in Zelle E4 der Arbeitsbeginn vom Pausenbeginn und das Pausenende vom Arbeitsende abgezogen, und die jeweiligen Ergebnisse werden addiert:

$$=(B4-A4)+(D4-C4).$$
- ✓ In Zelle F7 wird der Bereich E4:E6 mit der Funktion SUMME addiert. Standardmäßig beginnt die Anzeige nach 24 Stunden wieder bei 0. Damit die Zelle die addierte Zeit über einen Zeitraum von mehr als 24 Stunden anzeigen kann, wählen Sie im Dialogfenster ZELLEN FORMATIEREN (Register ZAHLEN) im Listenfeld KATEGORIE den Eintrag UHRZEIT und im Listenfeld TYP das Uhrzeitformat 37:30:55.

Beispiel: Altersgrenzen berechnen

Sie wollen eine Jugendfahrt organisieren. Da auch Minderjährige mitfahren möchten, benötigen Sie für die minderjährigen Teilnehmer eine Einverständniserklärung der Eltern. Um schnell zu erkennen, von wem eine solche Bescheinigung notwendig ist, berechnen Sie das Alter der Jugendlichen.

Das momentane Alter lässt sich in Excel mit der Funktion DATEDIF berechnen. Die Funktion DATEDIF ermittelt die Differenz zwischen zwei Datumsangaben.

Syntax der Funktion:

DATEDIF(Ausgangsdatum;Enddatum;Einheit)

Mithilfe des Arguments *Einheit* können Sie festlegen, in welchen Zeiteinheiten die Differenz ausgegeben wird:

Argument <i>Einheit</i>	Ergebnis
"Y"	Die Anzahl der vollständigen Jahre in der angegebenen Zeitspanne wird ausgegeben.
"M"	Die Anzahl der vollständigen Monate in der angegebenen Zeitspanne wird ausgegeben.
"D"	Die Anzahl der Tage in der angegebenen Zeitspanne wird ausgegeben.

Die Funktion DATEDIF muss manuell eingegeben werden, da sie nicht über den Funktions-Assistenten eingegeben werden kann.

	A	B	C	D	E
1	Name	Geburtstag	Alter		
2	Weber	20.03.1989	17	Bescheinigung notwendig	
3	Binsen	03.11.1988	18	Ok	
4	Hark	01.05.1986	20	Ok	
5	Braun	14.12.1988	17	Bescheinigung notwendig	
6	Seeler	31.10.1986	20	Ok	
7	Maurer	10.07.1990	16	Bescheinigung notwendig	
8	Wessing	21.10.1987	19	Ok	
9	Thomann	10.03.1987	19	Ok	
10	Galan	01.07.1988	18	Ok	
11	Knepel	28.01.1988	18	Ok	

Volljährigkeit berechnen (aktuelles Datum: 03.11.2006)

- ✓ In Spalte B stehen die Geburtstage der Jugendlichen.
- ✓ Die Formel in Zelle C2 lautet: $=DATEDIF(B2;HEUTE();"Y")$
- ✓ Mit der Formel in Spalte D fragen Sie ab, ob das Alter in Spalte C unter 18 Jahren liegt. Sind die Jugendlichen volljährig, soll die Meldung *Ok* ausgegeben werden, andernfalls die Meldung *Bescheinigung notwendig*.
- ✓ Die Formel in Zelle D2 lautet: $=WENN(C2<18;"Bescheinigung notwendig";"Ok")$